

Fisica Sandbox: Lets Experiment!

Get Started: First, make sure you watch the [introductory video on Fisica](#). Then, copy and paste the Fisica experiment code at the end of the document into a new Processing sketch. Install the Fisica Library in Processing, and make sure you can run the code. Read over the code and interact with the running project. Play with it a bit, try to break things! Once you have played with it a bit, try the following tasks:

Tasks: I am assigning the following tasks to help you explore how Fisica works.

1. Change one of the ramps into a bucket-shaped FPoly that can catch the falling FBodies.
2. Using traditional animation techniques, write code to create two clouds that travel across the screen. When they reach the end of the screen, make them jump back to their starting place. Draw one of the clouds behind the FBodies, and the other one in front of the FBodies.
3. Change the green FBox objects into bouncy squares. Attach an image to the FBox the way that the red bird is attached to some of the FBodies in the starter code. Make sure to resize the image after loading it so that it fits perfectly on the FBox.
4. Create two buttons and place them on the screen in front of the FBodies. Clicking on one button should turn on/off the world's gravity. Clicking on the other button should turn on/off the adding of new FBodies to the world.

Here is the Fisica experiment code:

```
import fisica.*;

//palette
color blue    = color(29, 178, 242);
color brown   = color(166, 120, 24);
color green   = color(74, 163, 57);
color red     = color(224, 80, 61);
color yellow  = color(242, 215, 16);

//assets
PImage redBird;

FPoly topPlatform;
FPoly bottomPlatform;

//fisica
FWorld world;

void setup() {
  //make window
  size(800, 600);

  //load resources
  redBird = loadImage("red-bird.png");
```

```

//initialise world
makeWorld();

//add terrain to world
makeTopPlatform();
makeBottomPlatform();
}

//=====

void makeWorld() {
    Fisica.init(this);
    world = new FWorld();
    world.setGravity(0, 900);
}

//=====

void makeTopPlatform() {
    topPlatform = new FPoly();

    //plot the vertices of this platform
    topPlatform.vertex(-100, 60);
    topPlatform.vertex(640, 240);
    topPlatform.vertex(640, 340);
    topPlatform.vertex(-100, 160);

    // define properties
    topPlatform.setStatic(true);
    topPlatform.setFillColor(brown);
    topPlatform.setFriction(0.1);

    //put it in the world
    world.add(topPlatform);
}

//=====

void makeBottomPlatform() {
    bottomPlatform = new FPoly();

    //plot the vertices of this platform
    bottomPlatform.vertex(900, 360);
    bottomPlatform.vertex(300, 480);
    bottomPlatform.vertex(300, 580);
    bottomPlatform.vertex(900, 460);

    // define properties
    bottomPlatform.setStatic(true);
    bottomPlatform.setFillColor(brown);

```

```

bottomPlatform.setFriction(0);

//put it in the world
world.add(bottomPlatform);
}

//=====

void draw() {
    println("x: " + mouseX + " y: " + mouseY);
    background(blue);

    if (frameCount % 50 == 0) { //Every 20 frames ...
        makeCircle();
        makeBlob();
        makeBox();
        makeBird();
    }
    world.step(); //get box2D to calculate all the forces and new positions
    world.draw(); //ask box2D to convert this world to processing screen coordinates and draw
}

//=====

void makeCircle() {
    FCircle circle = new FCircle(50);
    circle.setPosition(random(100,width-100), -5);

    //set visuals
    circle.setStroke(0);
    circle.setStrokeWeight(2);
    circle.setFillColor(red);

    //set physical properties
    circle.setDensity(0.2);
    circle.setFriction(1);
    circle.setRestitution(1);

    //add to world
    world.add(circle);
}

//=====

void makeBlob() {
    FBlob blob = new FBlob();

    //set visuals
    blob.setAsCircle(random(100,width-100), -5, 50);

```

```

blob.setStroke(0);
blob.setStrokeWeight(2);
blob.setFillColor(yellow);

//set physical properties
blob.setDensity(0.2);
blob.setFriction(1);
blob.setRestitution(0.25);

//add to the world
world.add(blob);
}

//=====

void makeBox() {
    FBox box = new FBox(25, 100);
    box.setPosition(random(100,width-100), -5);

    //set visuals
    box.setStroke(0);
    box.setStrokeWeight(2);
    box.setFillColor(green);

    //set physical properties
    box.setDensity(0.2);
    box.setFriction(1);
    box.setRestitution(0.25);
    world.add(box);
}

//=====

void makeBird() {
    FCircle bird = new FCircle(48);
    bird.setPosition(random(100,width-100), -5);

    //set visuals
    bird.attachImage(redBird);

    //set physical properties
    bird.setDensity(0.8);
    bird.setFriction(1);
    bird.setRestitution(0.5);
    world.add(bird);
}

```